Betriebssystemtechnik

Adressräume: Trennung, Zugriff, Schutz

II. Systemaufruf

Wolfgang Schröder-Preikschat

3. Mai 2022



Gliederung


Rekapitulation Mehrebenenmaschinen Teilinterpretierung

Funktionale Hierarchie Analogie Abstraktion

Implementierung
Entvirtualisierung
Befehlsarten

Zusammenfassung








@ wosch


- Die Befehlssatzebene interpretiert das Maschinenprogramm befehlsweise,
- 2. setzt dessen Ausführung aus,
 - Ausnahmesituation
 - Programmunterbrechung startet das Betriebssystem und
- 3. interpretiert die Programme des Betriebssystems befehlsweise.

Folge von 3.: Ausführung von Betriebssystemprogrammen

- 4. Das *Betriebssystem interpretiert* das soeben oder zu einem früheren Zeitpunkt unterbrochene Maschinenprogramm¹ befehlsweise und
- instruiert die Befehlssatzebene, die Ausführung des/eines zuvor unterbrochenen Maschinenprogramms¹ wieder aufzunehmen.







- Ausführung eines Maschinenprogramms
- Auslösung eines Systemaufrufs durch den Prozessor
- Verzweigung zum Betriebssystem und Behandlung des Systemaufrufs
- Beendigung des Systemaufrufs
- Rückverzweigung zum Maschinenprogramm



Gliederung

Rekapitulation Mehrebenenmaschinen Teilinterpretierung


Funktionale Hierarchie Analogie Abstraktion

Implementierung
Entvirtualisierung
Befehlsarten
Ablaufkontoxt

Zusammenfassung



Prozedur- vs. Systemaufruf




Maschinenprogramm Aufrufer call Aufrufstumpf shift Aufrufzuteiler Systemfunktion

- Betriebssystem
- Systemaufruf als adressraumübergreifender Prozeduraufruf
 - verlagert (shift) die weitere Prozedurausführung ins Betriebssystem



Abstraktion von Betriebssystemabschottung



Ortstransparenz


- durch den Aufrufstumpf
 - für den Aufrufer
- durch den Aufrufzuteiler
 - für die Systemfunktion

Entkopplung

- des Maschinenprogramms
- von Programmen des Betriebssystems



Abstraktion von Maschinenprogrammabschottung



Ortstransparenz

- durch den Aufrufstumpf
 - für den Aufrufer
 - durch den Aufrufzuteiler
 - für die Systemfunktion

Entkopplung

- des Maschinenprogramms
- von Programmen des Betriebssystems

Zugriffstransparenz

- durch Zugriffsmethoden
 - für den Aufrufzuteiler
 - für die Systemfunktion



Standard ist die **synchrone Programmunterbrechung** (*trap*)

Ausnahme (exception) von der "normalen" Programmausführung

OS/360 svc, für System/360 und danach

Unix V6 ■ trap, für PDP 11

Windows ■ int \$0x2e

Linux ■ int \$0x80, für x86

swi, für ARM

t, für SPARC

MacOS • \$0xa, für m68k: A-traps, illegaler Operationskode²

■ int \$0x80, für x86

im Vergleich zum normalen Prozeduraufruf, sehr kostspielig (S. 27)

Avantgarde sind Ansätze, die im Grunde frei von Aufrufsemantik sind

der Fokus liegt auf **Moduswechsel**: sysenter/syscall (x86-64)



²Motorola verwendete Befehle beginnend mit 1111₂ (reserviert für 68881, FPU-Koprozessor) und 1010₂ niemals in Prozessoren der 68000-Familie.

Gliederung

Rekapitulation Mehrebenenmaschinen Teilinterpretierung

Funktionale Hierarchie Analogie Abstraktion

Implementierung
Entvirtualisierung
Befehlsarten
Ablaufkontext

Zusammenfassung



Ebene $_5 \mapsto \mathsf{Ebene}_4$

Systemaufruf als Konstrukt problemorientierter Programmiersprache

```
char buf[1];
  done = read(0, buf, sizeof(buf));
4
  seine semantisch äquivalente Umsetzung in Assembliersprache (x86)
  ■ gcc -0 -m32 -fomit-frame-pointer -fno-pic -S
  subl $16, %esp
                        ; dispose parameter block:
  pushl $1
                            input buffer: length (in bytes)
                            input buffer: address
  pushl $buf
  pushl $0
                            file descriptor: standard input
  call read
                        ; execute system function
  movl %eax, done
                        : save return code
```



1

int done:

\$28, %esp

; release parameter block

addl

Systemaufruf als Konstrukt der Maschinenprogrammebene:

```
read:
1
                            backup callee-save register
    pushl
          %ebx
    movl
          16(%esp), %edx; pass 3rd input parameter
3
    movl 12(%esp), %ecx; pass 2nd input parameter
4
         8(%esp), %ebx; pass 1st input parameter
    movl
          $3
                          ; perform system call and return
    scar
6
    popl
          %ebx
                          ; restore callee-save register
    ret
```

- problemspezifische Varianten, je nach Betriebssystembefehlsart:
 - Primitivbefehl (RISC-artig), im Beispiel hier (Linux-artig) und ff.
 - Anzahl der zu sichernden nichtflüchtigen (callee-save) Register
 - Hauptspeicher oder flüchtige (caller-save) Register als Sicherungspuffer
 - stapel- oder registerbasierte Parameterübergabe
 - rückkehrende oder rückkehrlose Interaktion mit dem Betriebssystem
 - Komplexbefehl (CISC-artig), vgl. auch S. 21



rückkehrender Systemaufruf mit zwei Eingabeparametern:

```
kill:
movl %ebx, %edx ; backup into caller-save register
movl 8(%esp), %ecx ; pass 2nd input parameter
movl 4(%esp), %ebx ; pass 1st input parameter
scar $37 ; perform system call and return
movl %edx, %ebx ; restore from caller-save register
ret
```

rückkehrloser Systemaufruf mit einem Eingabeparameter:

```
1 _exit:
2  movl 4(%esp), %ebx ; pass input parameter
3  sc $252  ; perform system call, no return
```

rückkehrender parameterloser Systemaufruf:

```
getpid:
scar $20; perform system call and return
ret
```

Absetzen des Systemaufrufs

```
1 .macro sc scn
2 movl \scn, %eax ; pass system call number
3 int $128 ; cause software interrupt
4 .endm
```

Systemaufruf und Fehlerbehandlung nach Rückkehr

```
.macro scar scn
                       ; perform system call and return
    sc \scn
2
    cmpl $-4095, %eax; check for system call error
    jb
         .s\@
                         normal operation, if applicable
4
    neg %eax
                         derive (positiv) error code
5
    movl %eax, errno
                       ; put aside for possibly reworking
6
    movl $-1, %eax
                       ; indicate failure of operation
  .s\@:
                        come here if error free
8
  .endm
```

- Platzhalter f
 ür den Fehlerkode (im Datensegment, .data)
 - .long errno



- Problem: Schutzdomänen überschreitende Ausnahmeauslösung
 - normale Funktionsergebnisse von ausnahmebedingten unterscheiden
 - eine für das gesamte Rechensystem effiziente Umsetzung durchsetzen
- Lösungen dazu hängen ab von Betriebssystem und Befehlssatzebene
 - Wertebereich für Funktionsergebnisse beschneiden (z. B. Linux)
 - Wert im Rückgaberegister (%eax) zeigt den Ausnahme- oder Normalfall an

```
v \in [-1, -4095] \Rightarrow v \text{ ist Fehlerkode} \geq 0 \text{xfffff000 (x86)}
sonst \Rightarrow v \text{ ist Funktionsergebnis} < 0 \text{xfffff000 (x86)}
```

- betriebssystemseitig einfach, sofern alle Funktionsergebnisse dazu passen
- Übertragsmerker (carry flag) im Statusregister setzen³
 - Stapelrahmen (stack frame) des Systemaufrufs so manipulieren, dass bei Rückkehr der Merker den Ausnahme- (1) oder Normalfall (0) anzeigt
 - betriebssystemseitig mit größerem Mehraufwand (overhead) verbunden
- als Befehlssatzebenenerweiterung wäre der Merkeransatz konsequent

⁰

³Jeder Merker zur Steuerung bedingter Sprünge eignet sich dafür.

```
scd:
1
      pushl
             %ebp
      pushl
             %edi
3
      pushl
             %esi
4
      pushl
             %edx
5
      pushl %ecx
6
      pushl %ebx
7
             $NJTE, %eax
      cmpl
8
      jae
             scd_fault
g
      call
             *jump_table(,%eax,4)
10
   scd_leave:
11
      popl
             %ebx
12
             %ecx
      popl
13
      popl
             %edx
14
      popl
             %esi
15
                                       2
      popl
             %edi
16
                                       3
17
      popl
             %ebp
      iret
18
```


system call dispatcher.

- 2–7 i Sicherung
 - ii Parametertransfer
- 8–9 Überprüfung
 - 10 Ausführung
- 12–17 Wiederherstellung
 - 18 Wiederaufnahme

Fehlerbehandlung

```
scd_fault:
  movl $-ENOSYS,%eax
  jmp scd_leave
```





Stapelaufbau⁴ nach Aufruf der Systemfunktion über die Sprungtabelle
call *jump_table(,%eax,4)



⁴x86 real-address mode

```
extern int sys_ni_syscall(void);
   extern int sys_exit(int);
2
   extern int sys_fork(void);
3
   extern int sys_read(int, void*, int);
4
   extern int sys_write(int, void*, int);
5
6
   . . .
   #define NJTE 326 /* number of jump table entries */
8
9
   int (*jump_table[NJTE])() = {
                                      /* opcode */
10
            sys_ni_syscall,
                                      /*
                                           0 */
11
            sys_exit,
                                      /*
                                           1 */
12
                                      /*
                                           2 */
13
           sys_fork,
                                      /*
                                           3 */
14
           sys_read,
                                      /*
                                           4 */
15
           sys_write,
16
            . . .
   };
17
```



@ wosch

```
asmlinkage
   ssize_t sys_read(unsigned fd, char *buf, size_t count) {
        ssize_t ret;
        struct file *file;
4
5
        ret = -EBADF;
6
        file = fget(fd);
        if (file) {
8
                                     asmlinkage
9
                                     Instruiert gcc, die Funktionsparameter
        }
10
                                     auf dem Stapel zu erwarten und nicht in
        return ret;
11
                                     Prozessorregistern.
```

```
1
  asmlinkage long sys_ni_syscall(void) {
      return -ENOSYS;
```



12

Primitivbefehl (x86)

```
movl op6, %ebp
  movl op5, %edi
  movl op4, %esi
  movl op3, %edx
  movl op2, %ecx
  movl op1, %ebx
  movl opc, %eax
  int
        $42
8
```

Beachte

- bei Primitivbefehlen erfolgt die Auswertung der Operanden dynamisch, zur Laufzeit
 - Prozessorregister müssen freigemacht werden
- bei Komplexbefehlen geschieht dies statisch, zur Assemblier-/Bindezeit, und registerlos
- Komplexbefehl (x86): uniforme (li.) oder individuelle (re.) Operanden

```
int $42
   .long opc
2
   .long op1
3
   .long op2
4
    further operands
   .long opn
```


```
int $42
.byte
       opc
.align 4
.long op1
.long
       op2
further operands/alignments
```

opn

.long

7


Parametertransfer: Primitivbefehl (Xunil $_{x86}$)



- Werteübergabe (call by value) für alle Parameter
 - Variable: Befehlsoperand ist Adresse im Datenspeicher inkl. Register
 - Direktwert: Bestandteil des Befehls im Befehlsstrom
- Systemaufrufe als Primitivbefehle sind (meist) Unterprogramme




@ wosch



- Systemaufrufparameter werden nicht (mehr) in Registern transferiert
 - Systemaufrufe sind Unterprogramme, Parameter werden gestapelt
 - in Ergänzung zum Registeransatz, falls die Parameteranzahl zu groß ist
- das Betriebssystem lädt Parameter direkt vom Benutzerstapel



@ wosch



- das Betriebssystem lädt Parameter direkt vom Benutzeradressraum
 - Werteübergabe (call by value) für alle Direktwerte
 - Referenzübergabe (call by reference) sonst
- Systemaufrufe als Komplexbefehle sind (meist) Makroanweisungen



- Primitivbefehl
 - +/- Werteübergabe von Operanden im Maschinenprogramm
 - +/- dynamische Operandenauswertung (Laufzeit)
 - durch Prozessorregistersatz begrenzte Operandenanzahl
 - betriebssystemseitig bestenfalls teilweise Zustandssicherung
 - maschinenprogrammseitiger Mehraufwand zum Operandenabruf
- Komplexbefehl
 - + entspricht dem (statischen) Befehlsformat der Befehlssatzebene
 - + kompakte Darstellung/Kodierung von Systemaufrufen
 - + vollständige betriebssystemseitige Zustandssicherung
 - $+/-\,$ statische Operandenauswertung (Assemblier- oder Bindezeit)
 - Referenzübergabe von Operanden im Maschinenprogramm
 - betriebssystemseitiger Mehraufwand zum Operandenabruf
- wie gravierend die Negativpunkte sind, hängt vom Anwendungsfall ab



Schutzdomänenwechsel

- reale Sicht: ursprünglicher Zweck von Systemaufrufen (um 1955)
 - transiente Maschinenprogramme und residente Systemsoftware trennen
- logische Sicht: Systemaufrufe aktivieren einen privilegierten Kontext
 - Abschottung des Betriebssystemadressraums
 - Wechsel hin zum eigenen Adressraum des Betriebssystems
 - Erweiterung um den Adressraum des aufrufenden Maschinenprogramms
 - Erlaubnis zur (eingeschränkten) Durchführung bevorrechtigter Funktionen
 - Speicher-/Geräteverwaltung, Ein-/Ausgabe, ..., Betriebssystemdienste
 - allgemein: direkte Ausführung von Programmen der Befehlssatzebene
 - Zusicherung eigener Softwarebetriebsmittel zur Programmausführung
 - Stapelspeicher: $1:1 \leadsto \text{prozessbasierter}$, $N:1 \leadsto \text{ereignisbasierter}$ Kern
 - Prozessorregistersatz: Sicherung/Wiederherstellung oder Spiegelung
- 17

Abschottung und bevorrechtigte Ausführung

Systemaufrufe als eine synchrone Programmunterbrechung (trap) zu realisieren, ist ein mögliches Mittel zum Zweck und kein Muss

- effektiv müssen mit dem Mittel zwei Eigenschaften durchsetzbar sein:
 - i **privilegierter Arbeitsmodus** für den Betriebssystemkern
 - ii $Integrit \ddot{a}t$ $Verhinderung\ einer\ Infiltration^5\ ersterer\ Eigenschaft$
- ein *Trap* ist hinreichendes Mittel, aber auch vergleichsweise teuer
 - Zustandssicherung, Speicher- bzw. Tabellensuchen (table look-up)

Systemaufrufbeschleunigung durch Spezialbefehle (Intel, Pentium II)

privilegierten Programmtext nahezu, in Reihe" (inline) anordnen: x86

privilegierten Programmtext nahezu "in Reihe" (inline) anordnen: x86

```
movl $1f, %edx ; user mode continuation address
movl %esp, %ecx ; user mode stack pointer
sysenter ; enlist in priviledged mode
1:
```

vgl. auch VDSO (virtual dynamic shared object) in Linux



Kontextwechsel der CPU ohne Kontextsicherung und Tabellensuche

sysenter

- setzt CS, EIP und SS, ESP auf systemspezifische Werte
- schaltet Segmentierung ab (CS und SS: $[0..2^{32} 1]$)
- sperrt asynchrone Programmunterbrechungen (IRQ)
- aktiviert Schutzring 0

sysexit

- setzt CS und SS auf prozessspezifische Werte
- setzt EIP/ESP auf die in EDX/ECX stehenden Werte
- aktiviert Schutzring 3 nur von Ring 0 aus ausführbar
- das Betriebssystem belegt modellspezifische Register der CPU vor
 - MSR (model-specific register) 174h, 175h, 176h: CS, ESP und EIP, resp.
 - bei sysenter: SS = MSR[174h] + 8
 - bei sysexit: CS = MSR[174h] + 16, SS = MSR[174h] + 24
 - lacktriangledown mit MSR[174h] als eine Art "Basisindexregister" in die Segmenttabelle
- Kontextsicherung liegt komplett in Hand des Benutzerprozesses...
- alternativ: syscall/sysret (ursprünglich AMD; aber auch Intel 64)



Zusicherung eigener Softwarebetriebsmittel

Prozessorregistersatz

- im Regelfall durch Sicherung und Wiederherstellung von Registerinhalten
 - etwa der Stapelzeiger bei x86 [2]: Tupel (SS, ESP) sichern⁶
 - Statusregister und Befehlszeiger (program counter) sichern
 - alle, nur flüchtige oder wirklich verwendete Arbeitsregister sichern [4]
 - → dazu den Stapelspeicher des Betriebssystemkerns nutzen → Stapelwechsel
- verschiedentlich auch (zusätzlich) durch Spiegelung einzelner Register
 - etwa der Stapelzeiger beim MC68020: A7 \Leftrightarrow SP und USP [3]


Stapelspeicher

- dem Systemaufruf einen Stapel für den Betriebssystemkern zuteilen
 - → logische Konsequenz, wenn der Betriebssystemadressraum abgeschottet ist
 - einen Stapel im Betriebssystem für alle Kernfäden im Maschinenprogramm

 → typisch für ereignisbasierte Kerne (N:1)
 - einen Stapel im Betriebssystem pro Kernfaden im Maschinenprogramm \hookrightarrow typisch für prozessbasierte Kerne (1:1)
- ähnlich wird (oft) bei asynchronen Programmunterbrechungen verfahren




⁶Ausnahme *real-address mode*.



Prozessverdrängung/-blockierung im Kern ist (fast) überall möglich





Prozessverdrängung/-blockierung im Kern ist bedingt möglich [1]

Gliederung

Rekapitulation
Mehrebenenmaschinen
Teilinterpretierung

Funktionale Hierarchie Analogie Abstraktion

Implementierung
Entvirtualisierung
Befehlsarten
Ablaufkontoxt

Zusammenfassung



Resümee

- Rekapitulation
 - Maschinenprogramme werden durch Betriebssysteme teilinterpretiert
 - Teilinterpretierung wird (insb. auch) durch Systemaufrufe ausgelöst
- funktionale Hierarchie
 - Systemaufrufstümpfe trennen Maschinenprogramm von Betriebssystem
 - im Betriebssystem aktiviert ein Systemaufrufzuteiler die Systemfunktionen
 - der Systemaufruf ist ein adressraumübergreifender Prozeduraufruf
- Implementierung
 - ein Systemaufruf ist als Primitiv- oder Komplexbefehl realisiert
 - Primitivbefehle nutzen (ausschließlich) Register zur Parameterübergabe
 - Komplexbefehle erlauben einen unverfälschten Zustandsabzug
 - Fehler werden durch spezielle Rückgabewerte oder Merker signalisiert
 - ullet einem Systemaufruf ist ein Betriebssystemstapel 1:1 oder N:1 zugeteilt



Literaturverzeichnis I

[1] Draves, R.; Bershad, B. N.; Rashid, R. F.; Dean, R. W.:
Using Continuations to Implement Thread Management and Communication in
Operating Systems.

In: Proceedings of the Thirteenth ACM Symposium on Operating System Principles (SOSP 1991), ACM Press, 1991. – ISBN 0-89791-447-3. S. 122-136

- [2] INTEL CORPORATION (Hrsg.): Intel 64 and IA-32 Architectures: Software Developer's Manual. Order Number: 325462-045US. Santa Clara, California, USA: Intel Corporation, Jan. 2013
- [3] MOTOROLA SEMICONDUCTOR PRODUCTS INC. (Hrsg.): MC68020-MC68EC02009E Microprocessors User's Manual. First Edition. Phoenix, Arizona, USA: Motorola Semiconductor Products Inc., 1992
- [4] SCHRÖDER-PREIKSCHAT, W.; KLEINÖDER, J.: Systemprogrammierung. http://www4.informatik.uni-erlangen.de/Lehre/WS08/V_SP, 2008 ff.



Literaturverzeichnis II

[5] TANENBAUM, A. S.:


Multilevel Machines.

In: Structured Computer Organization.

Prentice-Hall, Inc., 1979. -


ISBN 0-130-95990-1, Kapitel 7, S. 344-386





- die Systemfunktion lädt Parameter direkt vom Benutzerstapel
 - indirekte Adressierung durch einen Zeiger auf den Parameterblock
 - Verzicht auf Ortstransparenz in der Systemfunktion
- der Prozessorstatus ist komplett betriebssystemseitig gesichert





- Systemaufrufparameter indirekt über einen Kontextdeskriptor laden
 - den Parameterblock vom Benutzerstapelzeiger ableiten
 - unterstützt insb. die merkerbasierte Signalisierung von Fehlerkodes
- Offenlegung des durch die CPU gesicherten Prozessorzustands

